
1.1 Overview 
 
The VWDSP Interface provides the necessary excitation and signal processing to read 2 vibrating wire 
gages with their temperature devices. It uses analog and digital filtering techniques to insure a highly 
reliable vibrating wire measurement output. The temperature input is compatible with a wide range of 
thermistors and similar temperature sensors.  
 
In addition, the VWDSP includes multiplexer expansion capabilities, up to 8 multiplexers may be enabled 
through the Interface. The VWDSP can also clock multiplexers using its CLK output. The default mode is 
for the attached Control Module to provide the clocking signal to advance the multiplexer channels. 
 

 
 
Control of the VWDSP is performed using simple ASCII based commands, transmitted using a serial data 
interface at 1200 bps.  
 

1.2 Specifications 
 
General 
Power requirements:  11-16 VDC (unregulated), nominal 12 VDC 
Quiescent current (with serial port disconnected):  < 500 µA 
Quiescent current (with serial port connected):  < 2.9 mA* 
VWDSP activated current:  < 40 mA 
VWDSP (during VW Ping on Chan B) current:  < 90 mA 
Control line input impedance:  100 kilohms 
Control line input levels:  TTL (5V logic) or RS-232 (+/-12V NRZ logic) 
Mux Enable Output levels (including CLK): TTL (5V logic) (< 6mA each) 
Power input transient protection: 17.1 VDC, 1500W Transzorbs 
with reverse polarity protection 
Control signal input transient protection: 18 VDC, 1500W Transzorbs 
Operating temperature:  -40 to +70° C (-40 to +160° F) 
* Radio connected to serial port can disconnect during sleep mode. 
 
Vibrating Wire Sensor 
Channels: 2 (each with thermistor input) 
Sweep Range:  
VW Coil resistance: 20 ohms minimum 
Thermistor Input: YSI 44005 

2.1 Terminal Connections 
 
The connections for the VWDSP are shown as follows: 
 
 

 
 
The following table explains the connections. 
 
Terminal Group Description 
12V Control Module Nominal 12V power input (11-16VDC range) 
GND Control Module Ground connection 
EN Control Module Enable input  
RD Control Module Receive data input (output from MCU) 
TD Control Module Transmit data output (input to MCU) 
CLK Control Module Clock input from MCU (redirected to Multiplexers CLK terminals) 
ENMX1-
ENMX8 

Multiplexers Multiplexer enable outputs 

CLK, CLK Multiplexers Clock outputs, whether from MCU (if connected) or generated 
internally. Two outputs provide 2 parallel TTL drive capability.  

VW+  VW Gage A Vibrating Wire + connection for channel A. 
VW- VW Gage A Vibrating Wire - connection for channel A. 
TH+ VW Gage A Temperature + connection for channel A. 
TH- VW Gage A Temperature - connection for channel A. 
SHLD VW Gage A Shield connection for channel A. 
VW+  VW Gage B Vibrating Wire + connection for channel B. 
VW- VW Gage B Vibrating Wire - connection for channel B. 
TH+ VW Gage B Temperature + connection for channel B. 
TH- VW Gage B Temperature - connection for channel B. 
SHLD VW Gage B Shield connection for channel B. 

2.2 Controller Wiring 
 
The VWDSP Interface provides a TTL or RS-232 level interface for connecting to the host device. The 
default interface is RS-232, contact Canary Systems or your product vendor to configure the interface for 
TTL. The VWDSP does not have data storage capability so it must be used in conjunction with other 
devices to deploy a data acquisition system. Typically it is used in conjunction with control modules such 
as the Campbell Scientific CR10X, CR800 or CR1000. 
 
The following table lists the Control Module and RS-232 connections, brief descriptions and connection 
destinations for use with a PC, and for the Control Modules, in both Non-MultiSensor Interface and 
MultiSensor Interface configurations. The MultiSensor Interface is a device manufactured by Canary 
Systems which facilitates connecting various types of instruments to a single multiplexer. 
 
Label Description CR10X 

Non-
MultiSensor 

CR10X 
MultiSensor 

CR800/ 
CR850 

CR1000 RS-232 
Female 

DB-9 
12V Power supply input. Nominal voltage 

is 12V DC, range is 6V to 16V. 
12V 12V 12V 12V  

GND Power supply ground. G G G G 5 (Black) 
EN Enable input. If the VWDSP must be 

powered continuously then connect 
to 12V, otherwise connect to a 
digital control port to activate. Input 
range is 4V to VCC (power supply). 
DO NOT APPLY A HIGHER 
VOLTAGE THAN THE POWER 
SUPPLY! 

C6 C8 C3 C7 4 (Red) 

RD Receive data input to the VWDSP. 
Default is RS-232 type, 1200bps, 8 
data bits, 1 stop bit, no parity. 

C4 C2 C1 C5 3 (White) 

TD Transmit data output to the VWDSP. 
Default is RS-232 type, 1200bps, 8 
data bits, 1 stop bit, no parity. 

C5 C3 C2 C6 2 (Green) 

CLK Multiplexer clock input. This 
connection is common to the 
Multiplexers CLK outputs. It is 
designed to facilitate expanding the 
CLK output to multiple multiplexers. 

C8 
(Note: The 

VWDSP can also 
generate the 

clocking signal to 
avoid using C8.) 

NA C4 
(Note: The 

VWDSP can 
also generate 
the clocking 

signal to avoid 
using C4.) 

C8 
(Note: The 

VWDSP can 
also generate 
the clocking 

signal to 
avoid using 

C8.) 

 

 
2.3 Multiplexer Wiring 
 
The schematic illustrates wiring of the 
VWDSP in a non-MultiSensor system 
with 2 multiplexers. 

2.4 Command Interface 
 
The VWDSP supports a simple ASCII based command set to configure, read and output measurements 
from the 2 vibrating wire channels. All commands are followed by a carriage return. The commands, with 
parameters, are as follows: 
 
Cnnnn – Clock multiplexer, where n = 0001-0256. Alternately the C command can be issued without a 4 
digit value to send a single clock pulse. The clock pulse width is 10ms.  
 
Mn – Enable multiplexer n, where n = 1-8. The respective ENMX1-ENMX8 output port is activated. If there 
is an existing port enabled it will be disabled prior to enabling the new port. 
 
Pssss pppp cccc mmmm ttttt – Configure the vibrating wire measurement. 
 
Where; 
ssss = Start frequency of excitation, in Hertz. Range is 0001-9999. 
pppp = Stop frequency of excitation, in Hertz. Range is 0001-9999. 
cccc = Number of frequency cycles. Range is 0001-9999. Recommended default is 500. 
mmmm = Duration of sampling period in 0.01 seconds. Range is 0001-9999. Recommended default is 
100 for a 1 second sampling period.  
tttt = Width of frequency swath. Range is 0001-9999. Recommended default is 100. 
 
For example, the following command configures the vibrating wire measurement for a start frequency of 
400 Hertz, stop of 3500 Hertz, 500 excitation cycles, a 1 second sampling period and a swath of 100. 
 
P0400 3500 0500 0100 0100 
 
If the command is received properly then the VWDSP will respond with OK, otherwise NG will display. 
 
VA – Output the vibrating wire measurement from Channel A. The output is in a raw data format, it must 
be converted to “digits” or period using simple math functions. The output consists of 4 values, separated 
by the space character and followed by a checksum. For example, consider the following example output: 
 
*VA<CR> 
VA734 733 112 60579 3A 
* 
 
The first value, 734, is the total of available counts. 
The second value, 733, is the useable counts. 
The third value, 112, is the Sum MSW, followed by the Sum LSW, 60579. 
 
To convert these values to period multiply the Sum MSW (112) by 65536 and add to this result the Sum 
LSW (60579). Then divide the number of counts used into the Sum to derive counts per cycle, then 
multiply by the clock period, or 0.1356µs, for count period. For example, using the figures in the example 
above, the vibrating wire period is (((112*65536)+60579)/733)*0.1356 or 1369µs. Invert this to derive 
frequency, square and then multiply by 0.001 to derive digits. For example, (1/.001369)2*0.001 = 533 
digits. Divide useable counts into available counts to derive reading quality. 
 
See the CR10X and CR1000 program examples later in this section 
 
VB – Output the vibrating wire measurement from Channel B. See above for conversion to period or 
digits. 

TA – Output the temperature device measurement from Channel A. The output is in a raw data format, it 
must be converted to resistance or temperature using simple math functions. The output consists of 2 
values, separated by the space character and followed by a checksum. 
 
Note: The VWDSP Firmware was updated beginning with version 8 (use the S command to verify 
the version of your unit) to output different units for the TA command. The following 2 sections 
will clarify the different output units. 
 
Pre Version 8 Firmware 
 
Consider the following example output: 
 
*TA<CR> 
TA511 1014 94 
* 
 
The first value, 511, is the excitation voltage, expressed in terms of the 10-bit (0-1023) range of the A/D 
converter. The second value, 1014, is the output voltage, expressed in terms of the 10-bit (0-1023) range 
of the A/D converter. 
 
To convert these values to temperature first calculate the current used to power the temperature device. 
Divide the voltage input by 1023, multiply by the reference voltage and then divide by 1000 to derive the 
current. For example, ((511/1023)*4.775)/1000 = 0.002385A.  To derive the temperature device 
resistance subtract the input voltage from the output and divide by the current. For example, 
(((1014/1023)*4.775) - ((511/1023)*4.775))) /0.002385 = 984 ohms.  
 
Version 8 and Higher 
 
*TA<CR> or TAnnnn<CR> 
TA00000 63800 B1 
* 
 
The 2 values represent a MSW and LSW of the 10-bit A/D result of the voltage measurement. By default 
the VWDSP acquires 100 samples and sums the results. Use TAnnnn to specify alternate averaging, 
nnnn=number of samples to sum. The MSW represents the number of 65536 results, the LSW represents 
the remainder.  
 
To convert these values to temperature first calculate the resistance of the thermistor. This requires 4 
steps. First, calculate the voltage across the termination resistor (6040). Vtr = ((((0*65536)+63800) 
/100)/1023)*2.5 = 1.559 Second, calculate the current through the thermistor by dividing Vtr into the 
resistor value (6040). Ir = 1.559/6040 = .000258 Third, calculate the voltage across the current limiting 
resistor by multiplying the resistance (499) by the current (Ir). Vr = 499 * .000258 = .1288 Fourth, 
calculate the thermistor resistance by subtracting Vtr and Vr from the excitation voltage and then dividing 
by the current (Ir). R = (2.5 – Vtr – Vr)/.000258 = (2.5 – 1.559 - .1288)/.000258 = 3148 ohms.  
 
These formulas can be simplified but in their expanded form help explain the calculations performed. 
 
The next step for either output is to convert the resistance values to temperature. 

Conversion Resistance to Temperature 
 
The conversion from resistance to temperature will depend on the model of the temperature device. For 
example, a YSI44005 device would be converted to temperature using the following log equation: 
 

T
A B LnR C LnR


 


1

273 23( ) ( )
.  

Where; T  Temperature in C. 
 LnR  Natural Log of Thermistor Resistance 

 A  1.4051  10-3  

 B  2.369  10-4   

 C  1.019  10-7  
 
Using the example above the temperature is approximately 52.5°C for the <rev 8 firmware and 
approximately 24.0°C for the >= rev 8 firmware. 
 
TB – Output the temperature device measurement from Channel B. See above for conversion to 
resistance or temperature. 
 
2.5 MultiLogger Software Configuration 
 
Support for the VWDSP is included in MultiLogger beginning in version 2.0.12. Contact Canary Systems 
or your software vendor to obtain an update. You can find the version of your software by selecting the 
menu item Help | About.  
 
NOTE: There have been important updates to MultiLogger in regards the functionality of the 
VWDSP, contact Canary Systems or your software vendor to obtain the current version.  
 
There are 3 steps to the configuration of your system to use the VWDSP. 
 
1.Configure the multiplexer Gage Type  
 
If using the MultiSensor Interface then select MultiSensor, 
as shown. 
 

 
 
 
 
 
 
 
If NOT using the MultiSensor Interface then select VWDSP, 
as shown. 
 
The Gage Type settings shown relate to the wiring of your 
system. If you are not sure of the system wiring, or whether 
the system includes the MultiSensor Interface, then contact 
your hardware vendor for further direction. 

2. Configure the Channel Gage Type 
 
There are generic VWDSP gage types as well as model specific VWDSP gage types. It is highly 
recommended to use the model specific gage types wherever possible. Note how the selections are 
organized by Gage Type | Make | Model. The VWDSP types will have VWDSP in the Model name.  
 
Contact Canary Systems or your software vendor for information regarding gages not listed. 
 
The generic types are explained below. 
 

 
 
The generic models available include: 
 
VWDSP_REVB – Frequency range is 400 to 3500 Hz, digits output, uses channel A of the VWDSP.  
VWDSP_REVB_B – Frequency range is 400 to 3500 Hz, digits output, uses channel B of the VWDSP. 
VWDSP_REVBP – Frequency range is 400 to 3500 Hz, period output (in microseconds), uses channel A 
of the VWDSP. 
VWDSP_REVB_BP – Frequency range is 400 to 3500 Hz, period output (in microseconds), uses channel 
B of the VWDSP. 
 
Note: The model specific gage types were added to MultiLogger beginning with version 2.1.2, it is 
highly recommended that you upgrade your version of MultiLogger if it pre-dates this version. 
Contact Canary Systems or your software vendor to obtain the current version.  
 
IMPORTANT NOTE: All of the Canary Systems VWDSP gage types utilize UserLoc1 (Check your 
Input Storage list for the exact location number) to report the measurement Quality. This value 
expressed as a percentage, represents the amount of data that was used, compared to the amount 
of data available, in the calculation of digits or period. The Canary Systems VWDSP selections use 
a threshold of 50% to determine validity of a measurement. Be aware that other Application Notes 
may describe using this location to store various parameters related to alarm monitoring or other 
uses, please select an alternate location when utilizing the Canary Systems VWDSP gage types. 

3. Configure the Upper Channel (if 16 Channel multiplexer) 
 
If the multiplexer Channels is set to 16 then a temperature may also be read by making the appropriate 
selection from the Upper Channel Device list. If the multiplexer Channels is set to 32 or 48 then the 
Upper Channel options will be disabled. 
 
If using the MultiSensor Interface then select the appropriate Thermistor device, as shown below. 
 

 
 
If NOT using the MultiSensor Interface then select the appropriate VWDSP Device, as shown below. 
 

 
 
There are 4 VWDSP devices available: 
 
VWDSP_THERMA – YSI44005 thermistor connected to the A channel with output in °C. 
VWDSP_ THERMA°F – YSI44005 thermistor connected to the A channel with output in °F. 
VWDSP_THERMB - YSI44005 thermistor connected to the B channel with output in °C. 
VWDSP_THERMB°F – YSI44005 thermistor connected to the B channel with output in °F. 

2.6 CR10X Programming Examples 
 
If you are writing your own Campbell program instead of using MultiLogger the following code example, 
extracted from MultiLogger, serves to illustrate how to utilize the VWDSP. This example uses the non-
MultiSensor system wiring and assumes the multiplexers are wired to use the A Channel. Output is in 
digits.  
 
Some of the nomenclature that follows in this program example is unique to the MultiLogger construct of 
programming files. For example, the Loc references (Tdbyte1, Tdbyte2, etc.) should be converted to Input 
Storage location numbers (with an appropriate matching label) when using this code in a Campbell 
program. 
 
The P command that is sent matches the example shown in the section Command Interface. 
 
Consult the Campbell CR10X Operators Manual for additional information on the P15 instruction. 
 
NOTE: This instruction example is broken into 3 sections, the first section describes the multiplexer 
activation and looping instructions, the second describes the vibrating wire measurement, the third 
describes the temperature measurement. 
 
Turn On VWDSP and Clock Multiplexer Loop 
 
;Sample VWDSP loop instructions 
 
;Configure PORTS and turn on VWDSP 
P20 Set Port(s) ;  
1:[4919                ] C8,C7,C6,C5 Options ;  
2:[9999                ] C4,C3,C2,C1 Options ;  
P22 Excitation with Delay ;  
1:[01                  ] Ex Channel ;  
2:[0000                ] Delay W/Ex (units = 0.01 sec) ;  
3:[0010                ] Delay After Ex (units = 0.01 sec) ;  
4:[0000                ] mV Excitation ;  
 
;Load M command 
P30 Z=F ;  
1:[77.000              ] F ;  
2:[00                  ] Exponent of 10 ;  
3:[TDbyte1             ] Z Loc                                ] ;  
P30 Z=F ;  
1:[49.000              ] F ;  
2:[00                  ] Exponent of 10 ;  
3:[TDbyte2             ] Z Loc                                ] ;  
P30 Z=F ;  
1:[13.000              ] F ;  
2:[00                  ] Exponent of 10 ;  
3:[TDbyte3             ] Z Loc                                ] ;  
 
;Send M Command 
P15 Port Serial I/O ;  
1:[01                  ] Reps ;  
2:[01                  ] P15 SERIAL I/0 (ASCII/RS-232, 1200 Baud) ;  
3:[0001                ] P15 Serial I/0 (Delay (units = 0.01 sec)) ;  
4:[03                  ] First Control Port ;  
5:[TDbyte1             ] Output Loc                    ] ;  
6:[0003                ] No. of Locs to Send ;  
7:[0000                ] Termination Character ;  
8:[0000                ] Maximum Characters ;  
9:[0000                ] CTS/Input Wait (units = 0.01 sec) ;  
10:[TDbyte1             ] Loc                    ] ;  
11:[1.0000              ] Mult ;  
12:[0.0000              ] Offset ;  
 
P22 Excitation with Delay ;  
1:[01                  ] Ex Channel ;  
2:[0000                ] Delay W/Ex (units = 0.01 sec) ;  
3:[0010                ] Delay After Ex (units = 0.01 sec) ;  
4:[0000                ] mV Excitation ;  

;Mux read loop 
P87 Beginning of Loop ;  
1:[0                   ] Delay ;  
2:[16                  ] Loop Count ;  
 
;Advance the channel 
P86 Do ;  
1:[78                  ] Command Code Option (Pulse Port 8) ;  
 
;*********************************************************************** 
;    CR10X Programming – Vibrating Wire Channel A Instructions Here 
;*********************************************************************** 
 
P31 Z=X ;  
1:[ReadingLoc          ] X Loc ;  
2:[Mux1Loc--           ] Z Loc ;  
 
P95 End ;  
 
;Turn off VWDSP 
P86 Do ;  
1:[56                  ] Command Code Option (Set Port 6 Low) ;  

 
Vibrating Wire Channel A  
;Load P Command 
P30 Z=F ;  
1:[80                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte1             ] Z Loc ;  
 
;Start 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte2             ] Z Loc ;  
 
P30 Z=F ;  
1:[52                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte3             ] Z Loc ;  
 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte4             ] Z Loc ;  
 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte5             ] Z Loc ;  
 
P30 Z=F ;  
1:[32                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte6             ] Z Loc ;  
 
;Stop 
P30 Z=F ;  
1:[51                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte7             ] Z Loc ;  
P30 Z=F ;  
1:[53                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte8             ] Z Loc ;  
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte9             ] Z Loc ;  
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte10            ] Z Loc ;  

P30 Z=F ;  
1:[32                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte11            ] Z Loc ;  
 
;Cycles 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte12            ] Z Loc ;  
 
P30 Z=F ;  
1:[54                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte13            ] Z Loc ;  
 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte14            ] Z Loc ;  
 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte15            ] Z Loc ;  
 
P30 Z=F ;  
1:[32                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte16            ] Z Loc ;  
 
;Sampling 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte17            ] Z Loc ;  
 
P30 Z=F ;  
1:[49                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte18            ] Z Loc ;  
 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte19            ] Z Loc ;  
 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte20            ] Z Loc ;  
 
P30 Z=F ;  
1:[32                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte21            ] Z Loc ;  
 
;Swath 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte22            ] Z Loc ;  

P30 Z=F ;  
1:[49                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte23            ] Z Loc ;  
 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte24            ] Z Loc ;  
 
P30 Z=F ;  
1:[48                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte25            ] Z Loc ;  
 
;Terminator 
P30 Z=F ;  
1:[13                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte26            ] Z Loc ;  
 
;Load Read Command 
P30 Z=F ;  
1:[86                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte27            ] Z Loc ;  
 
P30 Z=F ;  
1:[65                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte28            ] Z Loc ;  
 
P30 Z=F ;  
1:[13                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte29            ] Z Loc ;  
 
;Load S command 
P30 Z=F ;  
1:[83                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte30            ] Z Loc ;  
 
P30 Z=F ;  
1:[13                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte31            ] Z Loc ;  
 
;Send S command and get results 
P15 Port Serial I/O ;  
1:[1                   ] Reps ;  
2:[1                   ] P15:2 Serial I/0 (8-Bit, RS-232 ASCII, 1200 Baud) ;  
3:[1                   ] P15:3 Serial I/0 (Delay (0.01 sec units) before TX) ;  
4:[3                   ] First (RTS/DTR) of Control Ports Used ;  
5:[TDbyte30            ] Start Loc for TX ;  
6:[2                   ] Number of Locs to TX ;  
7:[42                  ] Termination Character for RX ;  
8:[5                   ] Max Characters to RX ;  
9:[100                 ] P15:9 Serial I/0 (Time Out for CTS (TX) and/or RX (0.01 sec 
units));  
10:[ScratchLoc1         ] Start Loc for RX ;  
11:[1                   ] Mult for RX ;  
12:[0                   ] Offset for RX ;  
 
P63 Extended Parameters ;  
1:[83                  ] Option ;  
2:[0                   ] Option ;  
3:[0                   ] Option ;  
4:[0                   ] Option ;  
5:[0                   ] Option ;  
6:[0                   ] Option ;  
7:[0                   ] Option ;  
8:[0                   ] Option ;  

;Make sure VWDSP responded before proceeding 
P89 If (X<=>F) ;  
1:[ScratchLoc1         ] X Loc ;  
2:[3                   ] Comparison Code Option (>=) ;  
3:[0                   ] F ;  
4:[30                  ] Command Code Option (Then Do) ;  
 
;Short delay before P command 
P22 Excitation with Delay ;  
1:[1                   ] Ex Channel ;  
2:[0                   ] Delay W/Ex (units = 0.01 sec) ;  
3:[10                  ] Delay After Ex (units = 0.01 sec) ;  
4:[0                   ] mV Excitation ;  
 
;Send P command 
P15 Port Serial I/O ;  
1:[1                   ] Reps ;  
2:[1                   ] P15:2 Serial I/0 (8-Bit, RS-232 ASCII, 1200 Baud) ;  
3:[1                   ] P15:3 Serial I/0 (Delay (0.01 sec units) before TX) ;  
4:[3                   ] First (RTS/DTR) of Control Ports Used ;  
5:[TDbyte1             ] Start Loc for TX ;  
6:[26                  ] Number of Locs to TX ;  
7:[0                   ] Termination Character for RX ;  
8:[0                   ] Max Characters to RX ;  
9:[0                   ] P15:9 Serial I/0 ;  
10:[TDbyte1             ] Start Loc for RX ;  
11:[1                   ] Mult for RX ;  
12:[0                   ] Offset for RX ;  
 
P22 Excitation with Delay ;  
1:[1                   ] Ex Channel ;  
2:[0                   ] Delay W/Ex (units = 0.01 sec) ;  
3:[10                  ] Delay After Ex (units = 0.01 sec) ;  
4:[0                   ] mV Excitation ;  
 
;Send VA command and get results 
P15 Port Serial I/O ;  
1:[1                   ] Reps ;  
2:[1                   ] P15:2 Serial I/0 (8-Bit, RS-232 ASCII, 1200 Baud) ;  
3:[10                  ] P15:3 Serial I/0 (Delay (0.01 sec units) before TX) ;  
4:[3                   ] First (RTS/DTR) of Control Ports Used ;  
5:[TDbyte27            ] Start Loc for TX ;  
6:[3                   ] Number of Locs to TX ;  
7:[42                  ] Termination Character for RX ;  
8:[32                  ] Max Characters to RX ;  
9:[1500                ] P15:9 Serial I/0 (Time Out for CTS (TX) and/or RX (0.01 sec units))
 ;  
10:[ScratchLoc1         ] Start Loc for RX ;  
11:[1                   ] Mult for RX ;  
12:[0                   ] Offset for RX ;  
 
P63 Extended Parameters ;  
1:[86                  ] Option ;  
2:[65                  ] Option ;  
3:[0                   ] Option ;  
4:[0                   ] Option ;  
5:[0                   ] Option ;  
6:[0                   ] Option ;  
7:[0                   ] Option ;  
8:[0                   ] Option ;  
 
;Make sure we have data from unit 
P89 If (X<=>F) ;  
1:[ScratchLoc1         ] X Loc ;  
2:[1                   ] Comparison Code Option (=) ;  
3:[-99999              ] F ;  
4:[30                  ] Command Code Option (Then Do) ;  
 
P31 Z=X ;  
1:[ScratchLoc1         ] X Loc ;  
2:[ReadingLoc          ] Z Loc ;  
 
P94 Else ;  

;Calculate Result 
P37 Z=X*F ;  
1:[ScratchLoc3         ] X Loc ;  
2:[65536               ] F ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P33 Z=X+Y ;  
1:[ScratchLoc4         ] X Loc ;  
2:[ScratchLoc5         ] Y Loc ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P38 Z=X/Y ;  
1:[ScratchLoc5         ] X Loc ;  
2:[ScratchLoc2         ] Y Loc ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc5         ] X Loc ;  
2:[0.1356              ] F ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P42 Z=1/X ;  
1:[ScratchLoc5         ] X Loc ;  
2:[ScratchLoc5         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc5         ] X Loc ;  
2:[1000                ] F ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc5         ] X Loc ;  
2:[1000                ] F ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P36 Z=X*Y ;  
1:[ScratchLoc5         ] X Loc ;  
2:[ScratchLoc5         ] Y Loc ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc5         ] X Loc ;  
2:[0.001               ] F ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc5         ] X Loc ;  
2:[0.001               ] F ;  
3:[ReadingLoc          ] Z Loc ;  
 
;Calculate Goodness 
P38 Z=X/Y ;  
1:[ScratchLoc2         ] X Loc ;  
2:[ScratchLoc1         ] Y Loc ;  
3:[ScratchLoc6         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc6         ] X Loc ;  
2:[100                 ] F ;  
3:[UserLoc1            ] Z Loc ;  

;Check Quality 
P89 If (X<=>F) ;  
1:[UserLoc1            ] X Loc ;  
2:[4                   ] Comparison Code Option (<) ;  
3:[50                  ] F ;  
4:[30                  ] Command Code Option (Then Do) ;  
 
P30 Z=F ;  
1:[-99998              ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[ReadingLoc          ] Z Loc ;  
 
P94 Else ;  
 
P89 If (X<=>F) ;  
1:[ScratchLoc2         ] X Loc ;  
2:[4                   ] Comparison Code Option (<) ;  
3:[100                 ] F ;  
4:[30                  ] Command Code Option (Then Do) ;  
 
P30 Z=F ;  
1:[-99997              ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[ReadingLoc          ] Z Loc ;  
 
P95 End ;  
 
P95 End ;  
 
P95 End ;  
 
;Else VWDSP is not responding 
P94 Else ;  
 
P31 Z=X ;  
1:[ScratchLoc1         ] X Loc ;  
2:[ReadingLoc          ] Z Loc ;  
 
P95 End ;  

CR10X Temperature 
 
The following code example illustrates how to read a YSI44005 type thermistor using the A channel of the 
VWDSP, with output in °C. 
P30 Z=F ;  
1:[83                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte1             ] Z Loc ;  
P30 Z=F ;  
1:[13                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte2             ] Z Loc ;  
 
;Load Read Command 
P30 Z=F ;  
1:[84                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte3             ] Z Loc ;  
P30 Z=F ;  
1:[65                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte4             ] Z Loc ;  
P30 Z=F ;  
1:[13                  ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[TDbyte5             ] Z Loc ;  
 
;Send S command and get results  
P15 Port Serial I/O ;  
1:[1                   ] Reps ;  
2:[1                   ] P15:2 Serial I/0 (8-Bit, RS-232 ASCII, 1200 Baud) ;  
3:[10                  ] P15:3 Serial I/0 (Delay (0.01 sec units) before TX) ;  
4:[3                   ] First (RTS/DTR) of Control Ports Used ;  
5:[TDbyte1             ] Start Loc for TX ;  
6:[2                   ] Number of Locs to TX ;  
7:[42                  ] Termination Character for RX ;  
8:[8                   ] Max Characters to RX ;  
9:[50                  ] P15:9 Serial I/0 (Time Out for CTS (TX) and/or RX (0.01 sec units))  
10:[ScratchLoc10        ] Start Loc for RX ;  
11:[1                   ] Mult for RX ;  
12:[0                   ] Offset for RX ;  
 
;Use S filter string 
P63 Extended Parameters ;  
1:[83                  ] Option ;  
2:[0                   ] Option ;  
3:[0                   ] Option ;  
4:[0                   ] Option ;  
5:[0                   ] Option ;  
6:[0                   ] Option ;  
7:[0                   ] Option ;  
8:[0                   ] Option ;  
 
;Check for S command received before proceeding 
P89 If (X<=>F) ;  
1:[ScratchLoc10        ] X Loc ;  
2:[3                   ] Comparison Code Option (>=) ;  
3:[0                   ] F ;  
4:[30                  ] Command Code Option (Then Do) ;  
 
;Send TA command and get results 
P15 Port Serial I/O ;  
1:[1                   ] Reps ;  
2:[1                   ] P15:2 Serial I/0 (8-Bit, RS-232 ASCII, 1200 Baud) ;  
3:[10                  ] P15:3 Serial I/0 (Delay (0.01 sec units) before TX) ;  
4:[3                   ] First (RTS/DTR) of Control Ports Used ;  
5:[TDbyte3             ] Start Loc for TX ;  
6:[3                   ] Number of Locs to TX ;  
7:[42                  ] Termination Character for RX ;  
8:[20                  ] Max Characters to RX ;  
9:[200                 ] P15:9 Serial I/0 (Time Out for CTS (TX) and/or RX (0.01 sec units))  
10:[ScratchLoc1         ] Start Loc for RX ;  
11:[1                   ] Mult for RX ;  
12:[0                   ] Offset for RX ;  

;Use TA filter string 
P63 Extended Parameters ;  
1:[84                  ] Option ;  
2:[65                  ] Option ;  
3:[0                   ] Option ;  
4:[0                   ] Option ;  
5:[0                   ] Option ;  
6:[0                   ] Option ;  
7:[0                   ] Option ;  
8:[0                   ] Option ;  
 
;Make sure we have a result from VWDSP 
P89 If (X<=>F) ;  
1:[ScratchLoc1         ] X Loc ;  
2:[4                   ] Comparison Code Option (<) ;  
3:[0                   ] F ;  
4:[30                  ] Command Code Option (Then Do) ;  
 
P30 Z=F ;  
1:[-99.99              ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[ReadingLoc          ] Z Loc ;  
 
P94 Else ;  
 
;Calculate Result for < v7 FW 
P89 If (X<=>F) ;  
1:[ScratchLoc10        ] X Loc ;  
2:[4                   ] Comparison Code Option (<) ;  
3:[8                   ] F ;  
4:[30                  ] Command Code Option (Then Do) ;  
 
P30 Z=F ;  
1:[1023                ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[ScratchLoc3         ] Z Loc ;  
 
P38 Z=X/Y ;  
1:[ScratchLoc1         ] X Loc ;  
2:[ScratchLoc3         ] Y Loc ;  
3:[ScratchLoc4         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc4         ] X Loc ;  
2:[4.700               ] F ;  
3:[ScratchLoc4         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc4         ] X Loc ;  
2:[0.001               ] F ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P38 Z=X/Y ;  
1:[ScratchLoc2         ] X Loc ;  
2:[ScratchLoc3         ] Y Loc ;  
3:[ScratchLoc6         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc6         ] X Loc ;  
2:[4.700               ] F ;  
3:[ScratchLoc6         ] Z Loc ;  
 
P35 Z=X-Y ;  
1:[ScratchLoc6         ] X Loc ;  
2:[ScratchLoc4         ] Y Loc ;  
3:[ScratchLoc7         ] Z Loc ;  
 
P38 Z=X/Y ;  
1:[ScratchLoc7         ] X Loc ;  
2:[ScratchLoc5         ] Y Loc ;  
3:[ScratchLoc7         ] Z Loc ;  
 
;Else calculate result for > v8 FW 
P94 Else ;  

;Multiple MSW 
P37 Z=X*F ;  
1:[ScratchLoc1         ] X Loc ;  
2:[65536               ] F ;  
3:[ScratchLoc1         ] Z Loc ;  
 
;Add our LSW 
P33 Z=X+Y ;  
1:[ScratchLoc1         ] X Loc ;  
2:[ScratchLoc2         ] Y Loc ;  
3:[ScratchLoc3         ] Z Loc ;  
 
;Divide for our averaging 
P37 Z=X*F ;  
1:[ScratchLoc3         ] X Loc ;  
2:[0.01                ] F ;  
3:[ScratchLoc3         ] Z Loc ;  
 
P30 Z=F ;  
1:[1023                ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[ScratchLoc4         ] Z Loc ;  
 
P38 Z=X/Y ;  
1:[ScratchLoc3         ] X Loc ;  
2:[ScratchLoc4         ] Y Loc ;  
3:[ScratchLoc4         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc4         ] X Loc ;  
2:[2.5                 ] F ;  
3:[ScratchLoc4         ] Z Loc ;  
 
P30 Z=F ;  
1:[6040                ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P38 Z=X/Y ;  
1:[ScratchLoc4         ] X Loc ;  
2:[ScratchLoc5         ] Y Loc ;  
3:[ScratchLoc5         ] Z Loc ;  
 
P37 Z=X*F ;  
1:[ScratchLoc5         ] X Loc ;  
2:[499                 ] F ;  
3:[ScratchLoc6         ] Z Loc ;  
 
P30 Z=F ;  
1:[2.5                 ] F ;  
2:[0                   ] Exponent of 10 ;  
3:[ScratchLoc7         ] Z Loc ;  
 
P35 Z=X-Y ;  
1:[ScratchLoc7         ] X Loc ;  
2:[ScratchLoc4         ] Y Loc ;  
3:[ScratchLoc7         ] Z Loc ;  
 
P35 Z=X-Y ;  
1:[ScratchLoc7         ] X Loc ;  
2:[ScratchLoc6         ] Y Loc ;  
3:[ScratchLoc7         ] Z Loc ;  
 
P38 Z=X/Y ;  
1:[ScratchLoc7         ] X Loc ;  
2:[ScratchLoc5         ] Y Loc ;  
3:[ScratchLoc7         ] Z Loc ;  
 
P95 End ;  
 
;Resistance should be in ScratchLoc7 
P40 Z=LN(X) ;  
1:[ScratchLoc7         ] X Loc ;  
2:[ScratchLoc8         ] Z Loc ;  

P30 Z=F ;  
1:[2.369               ] F ;  
2:[4--                 ] Exponent of 10 ;  
3:[ScratchLoc9         ] Z Loc ;  
 
P36 Z=X*Y ;  
1:[ScratchLoc8         ] X Loc ;  
2:[ScratchLoc9         ] Y Loc ;  
3:[ScratchLoc8         ] Z Loc ;  
 
P40 Z=LN(X) ;  
1:[ScratchLoc7         ] X Loc ;  
2:[ScratchLoc9         ] Z Loc ;  
 
P36 Z=X*Y ;  
1:[ScratchLoc9         ] X Loc ;  
2:[ScratchLoc9         ] Y Loc ;  
3:[ScratchLoc10        ] Z Loc ;  
 
P36 Z=X*Y ;  
1:[ScratchLoc9         ] X Loc ;  
2:[ScratchLoc10        ] Y Loc ;  
3:[ScratchLoc9         ] Z Loc ;  
 
P30 Z=F ;  
1:[1.019               ] F ;  
2:[7--                 ] Exponent of 10 ;  
3:[ScratchLoc10        ] Z Loc ;  
 
P36 Z=X*Y ;  
1:[ScratchLoc9         ] X Loc ;  
2:[ScratchLoc10        ] Y Loc ;  
3:[ScratchLoc9         ] Z Loc ;  
 
P30 Z=F ;  
1:[1.4051              ] F ;  
2:[3--                 ] Exponent of 10 ;  
3:[ScratchLoc10        ] Z Loc ;  
 
P33 Z=X+Y ;  
1:[ScratchLoc9         ] X Loc ;  
2:[ScratchLoc10        ] Y Loc ;  
3:[ScratchLoc10        ] Z Loc ;  
 
P33 Z=X+Y ;  
1:[ScratchLoc8         ] X Loc ;  
2:[ScratchLoc10        ] Y Loc ;  
3:[ScratchLoc10        ] Z Loc ;  
 
P42 Z=1/X ;  
1:[ScratchLoc10        ] X Loc ;  
2:[ScratchLoc10        ] Z Loc ;  
 
P34 Z=X+F ;  
1:[ScratchLoc10        ] X Loc ;  
2:[273.2--             ] F ;  
3:[ReadingLoc          ] Z Loc ;  
 
P95 End ;  
 
;No S command received - no VWDSP connected? 
P94 Else ;  
 
P31 Z=X ;  
1:[ScratchLoc10        ] X Loc ;  
2:[ReadingLoc          ] Z Loc ;  
 
P95 End ;  

2.7 CR1000 Programming Examples 
 
If you are writing your own Campbell program instead of using MultiLogger the following code example, 
extracted from MultiLogger, serves to illustrate how to utilize the VWDSP. Output is in digits. 
 
Some of the nomenclature that follows in this program example is unique to the MultiLogger construct of 
programming files. You will need to DIM ScratchLoc(10) as well as define a string buffer, sInBuf, to store 
the response string from the VWDSP. 
 
'Turn on VWDSP Power 
PortSet (7,1) 
'Wait for power up 
Delay(0,100,MSEC) 
'Open our serial port 
SerialOpen (8,1200,0,1000,255) 
 
'Check that the VWDSP is present by sending S command 
Delay (0,10,mSec) 
ScratchLoc(1) = SerialOut (8,"S"+CHR(13),"*",2,50) 
 
'Check for valid response 
if ScratchLoc(1) <> 0 then 
 
 'Short delay 
 Delay (0,100,mSec) 
 'Send P command 
 SerialOut (8,"P0400 3500 0600 0040 0300"+CHR(13),"",0,0) 
 'Short delay 
 Delay (0,350,mSec) 
 'Make sure buffer is clear 
 SerialFlush(8) 
 'Send VA command  
 SerialOut (8,"VA"+CHR(13),"",0,0) 
 'Receive response 
 SerialIn(sInBuf,8,1500,-1,30)  
   
 if Len(sInBuf) >= 30 then  
  'Split out response values 
  Splitstr(ScratchLoc(),sInBuf," ",4,0) 
   
  'Convert to reading 
  ScratchLoc(5) = 1/((((ScratchLoc(3)*65536)+ScratchLoc(4))/ScratchLoc(2))*0.1356) 
  ScratchLoc(5) = (ScratchLoc(5) * 1000000)^2 
  ScratchLoc(5) = ScratchLoc(5) * 0.001 'Convert to Digits otherwise f^2x10-9 
  Reading_Loc = ScratchLoc(5)  * 0.001 
 
  'Calculate quality 
  mlUser1 = (ScratchLoc(2) / ScratchLoc(1)) * 100 
 
  'Error - poor quality 
  if mlUser1 < 50 then Reading_Loc = -99.998   
   
  'Error - not enough data 
  if ScratchLoc(2) < 50 then Reading_Loc = -99.997 
  
 'No valid response 
 Else 
  Reading_Loc = -99.999 
 EndIf 
 
'No valid response 
Else 
 Reading_Loc = -99.999 
EndIf 
'Turn off VWDSP Power 
PortSet (7,0) 
'Close our serial port 
SerialClose (8) 

CR1000 Temperature 
 
The following code example illustrates how to program a CR1000 to read a YSI44005 type thermistor 
connected to the input channel A of the VWDSP. The firmware revision is checked for temperature 
conversion method. 
 
'Read a YSI44005 thermistor with the VWDSP Channel A - output C 
 
'Turn on VWDSP Power 
PortSet (7,1) 
'Wait for power up 
Delay(0,100,MSEC) 
'Open our serial port 
SerialOpen (8,1200,0,1000,255) 
'Short delay before commands 
Delay (0,50,mSec) 
 
'Check our VWDSP version to determine results processing 
SerialFlush(8) 
SerialOut(8,"S"+CHR(13),"",0,0) 
SerialIn(sInBuf,8,50,"*",12) 
SplitStr(ScratchLoc(10),sInBuf," ",2,0) 
 
'Make sure we have version before proceeding 
if ScratchLoc(10) > 0 then 
 
 SerialFlush(8) 
 SerialOut (8,"TA"+CHR(13),"",0,0) 
 SerialIn(sInBuf,8,100,-1,18)  
     
 if Len(sInBuf) > 16 then  
 
  SplitStr(ScratchLoc(),sInBuf," ",2,0)   
  
  'Check FW to determine how to convert results 
  if ScratchLoc(10) < 8 then 
   'Convert to resistance 
   ScratchLoc(4) = (ScratchLoc(1) / 1023) * 4.775 
   ScratchLoc(5) = ScratchLoc(4) * 0.001 
   ScratchLoc(6) = (ScratchLoc(2) / 1023) * 4.775 
   ScratchLoc(7) = (ScratchLoc(6) - ScratchLoc(4))/ScratchLoc(5) 
   else 
   ScratchLoc(3) = ((ScratchLoc(1) * 65536) + ScratchLoc(2)) / 100  
  
   ScratchLoc(4) = ((ScratchLoc(3)/1023)*2.5) 
   ScratchLoc(5) = ScratchLoc(4) / 6040 
   ScratchLoc(6) = ScratchLoc(5) * 499 
   ScratchLoc(7) = (2.5 - ScratchLoc(4) - ScratchLoc(6)) / ScratchLoc(5) 
   endif 
 
  'Use Steinhart/Hart to convert resistance to temperature 
  mlReading = 1/(.0014051+(.0002369*Log(ScratchLoc(7))) + 
(.0000001019*(Log(ScratchLoc(7))^3))) - 273.2 
 
  'Check range before finishing 
  if mlReading > 100 then mlReading = -99.8 
   
  else 
  mlReading = -99.9 
  endif 
 
 else 
 mlReading = -99999 
 endif 
 
'Turn off VWDSP Power 
PortSet (7,0) 
'Close our serial port 
SerialClose (8) 

3.1 Troubleshooting Flowchart 
 
If you cannot obtain readings using the VWDSP or the readings are unstable then see the troubleshooting 
flowchart below for help in determining the nature of the problem. If all fails contact Canary Systems by 
phone, fax or e-mail for further assistance. 
 

 

Does the VWDSP 
Ping the VW 

Sensor? 

Is 12V and G 
connected to the 

VWDSP? 

Are the sensor leads 
connected? 

Call Canary Systems for 
further assistance 

Is the 
wiring correct for  the 

sensor leads? 

Yes 

No 

Yes 

No Attach 12V and G 

Are the Enable and 
Control Connected 
to the control ports? 

Attach Enable & Control 
lines No 

Do the software 
settings for the control 

match the 
 serial port? 

Match software and 
digital Serial port settings 

Yes 

No 

Yes 

Yes 

Do the 
software settings match 

the sensor type ? 

Is there a source of 
extreme electrical 
   noise nearby? 

Yes 

No 

Connect the leads 

Adjust the wiring for the 
type of sensor used 

Adjust the software 
settings 

No 

No 

No 

Remove the noise 
source or move VWDSP Yes 

Start 

Does the 
VWDSP show 

water or other 
contamination? 

Yes Clean the VWDSP 
terminals  

No 

Yes 

 


